EMails mit Anhängen per VBA verschicken
Versionen: Access 2003, 2002/XP und 2000

Über die Helpline erreichen uns immer wieder Anfragen, rund um den Versand von E-Mails mit Anhängen aus Access. Mit diesem Tipp der Woche möchten wir eine Lösung vorstellen, die viele dieser Fragen beantwortet.

Beginnen wir aber mit einigen grundlegenden Anmerkungen: Access stellt mit "DoCmd.SendObject" zwar eine Funktion bereit, mit der Sie ein Objekt der aktuellen Datenbank als Anhang einer E-Mail versenden können, es dabei kann es sich immer nur ein um einzelnes Objekt pro E-Mailhandeln. Möchten Sie also zum Beispiel zwei Berichte "Produktübersicht" mit detaillierten Informationen zu Ihren Produkten und "Preisliste" mit den aktuellen Preisen an einen Kunden senden, geht das nur mit zwei Aufrufen von "DoCmd.SendObject", aus denen sich dann auch zwei E-Mails ergeben. Oftmals liegen die Produktinformationen auch als externes PDF-Dokument "Produktübersicht.pdf" vor. Soll nun dieses PDF-Dokument gemeinsam mit einem Bericht "Aktuelle Preisliste" an einen Kunden senden, ist das nicht möglich, da mit "DoCmd.SendObject" nur interne Datenbankobjekte aber keine externen Dateien als Anhang versendet werden können.

Die Lösung: Steuern Sie Outlook direkt per Office Automation an, erstellen Sie eine neue Nachricht und ordnen Sie dann neben Empfänger und Nachrichtentext beliebig viele Anhänge zu. Dazu ist es jedoch zunächst notwendig, die zu internen Datenbankobjekte in externe Dateien zu exportieren. Dazu verwenden Sie "DoCmd.OutputTo" beispielsweise wie folgt:

DoCmd.OutputTo acOutputReport, _
      "Produktübersicht", acFormatRTF, _
      "Z:\Export\Produktübersicht.rtf"
DoCmd.OutputTo acOutputReport, _
      "Preisliste", acFormatRTF, _
      "Z:\Export\Preisliste.rtf"

Diese Anweisungen exportieren die Berichte "Produktübersicht" und "Preisliste" im RTF-Format mit allen Formatierungen in ein externe Dateien "Produktübersicht.rtf" und "Preisliste.rtf" in das Verzeichnis "Z:\Export". Alternativ können Sie dem Kunden auch eine Excel-Arbeitsmappe mit den aktuellen Preisen schicken:

DoCmd.OutputTo acOutputTable, _
      "Artikel", acFormatXLS, _
      "Z:\Export\Artikel.xls"

Mit dieser Anweisung wird die komplette Tabelle "Artikel" als Excel-Arbeitsmappe "Artikel.xls" in das Verzeichnis "Z:\Export" exportiert. Möchten Sie nur Teile Ihrer Artikeldaten wie zum Beispiel Artikelnummer, Artikelname und Einzelpreis übermitteln, erstellen Sie zunächst eine auf Abfrage wie "Artikelauszug", die nur die gewünschten Felder selektiert und exportieren dann das Ergebnis der Abfrage als Excel-Arbeitsmappe:

DoCmd.OutputTo acOutputQuery, _
      "Artikelauszug", acFormatXLS, _
      "Z:\Test\Artikelpreise.xls"

Diese Anweisungen exportieren die über die Abfrage "Artikelauszug" selektierten Datensätze mit den gewünschten Feldern in die Excel-Arbeitsmappe "Artikelpreise.xls" in das Verzeichnis "Z:\Export".

Anschließend kann es mit dem Versand über Outlook weitergehen:

1. Legen Sie in einem "Menüformular" oder einem beliebigen anderen Formular zum Beispiel eine Schaltfläche Produktinfos und Preisliste senden (Name= btnSendReports) an.

2. Stellen Sie die Eigenschaft "Beim Klicken" der Schaltfläche auf den Eintrag "[Ereignisprozedur]" und klicken Sie auf die Schaltfläche mit den drei Punkten, um den VBA-Editor zu öffnen.

3. Wählen Sie das Menü Extras-Verweise an und setzen Sie eine Referenz auf "Microsoft Outlook x.0 Object Libary", wobei "x" für "11" (Outlook 2003), "10" (Outlook 2002/XP) oder "9" (Outlook 2000) steht.

4. Geben Sie im VBA-Editor die folgenden Anweisungen ein:

Private Sub btnSendReports_Click()
  Dim objOutl As Outlook.Application
  Dim objMail As Outlook.MailItem

  Set objOutl = CreateObject("Outlook.Application")
  Set objMail = objOutl.CreateItem(olMailItem)
  With objMail
    .Attachments.Add "Z:\Export\Produktübersicht.rtf"
    .Attachments.Add "Z:\Test\Preisliste.rtf"
    .Display
  End With

End Sub

Über "CreateObject()" wird eine Verbindung zu Outlook hergestellt und dort mit der Anweisung "CreateItem(olMailItem)" eine neue Nachricht angelegt. Dieser Nachricht weisen wir die zuvor generierten RTF-Dateien über die Auflistung "Attachments" (Anhänge) bzw. deren Methode "Add" als Anhänge zu. Die Anweisung ".Display" sorgt dafür, dass die neue Nachricht in einem Fenster im Vordergrund angezeigt wird und ermöglicht so die Vervollständigung der E-Mail um Empfänger und Nachrichtentext.

Selbstverständlich können Sie die gezeigte Ereignisprozedur auch mit den Anweisungen für den Export kombinieren:

Private Sub btnSendReports_Click()
  Dim objOutl As Outlook.Application
  Dim objMail As Outlook.MailItem

  DoCmd.OutputTo acOutputReport, _
      "Produktübersicht", acFormatRTF, _
      "Z:\Export\Produktübersicht.rtf"
  DoCmd.OutputTo acOutputReport, _
      "Preisliste", acFormatRTF, _
      "Z:\Export\Preisliste.rtf"

  Set objOutl = CreateObject("Outlook.Application")
  Set objMail = objOutl.CreateItem(olMailItem)
  With objMail
    .Attachments.Add "Z:\Export\Produktübersicht.rtf"
    .Attachments.Add "Z:\Export\Preisliste.rtf"
    .Display
  End With

End Sub

Die oben verwendeten Namen für Berichte und Dateinamen passen Sie entsprechend Ihrer Erfordernisse an und fügen gegebenenfalls weitere "DoCmd.OutputTo"- bzw. "Attachments.Add"-Anweisungen hinzu, um zum Beispiel andere externe Dateien als Anhang zuzuordnen:

Private Sub btnSendReports_Click()
  Dim objOutl As Outlook.Application
  Dim objMail As Outlook.MailItem

  DoCmd.OutputTo acOutputReport, _
      "Preisliste", acFormatRTF, _
      "Z:\Export\Preisliste.rtf"

  Set objOutl = CreateObject("Outlook.Application")
  Set objMail = objOutl.CreateItem(olMailItem)
  With objMail
    .Attachments.Add "Z:\Export\Preisliste.rtf"
    .Attachments.Add "Z:\Export\Produktübersicht.pdf"
    .Display
  End With

End Sub

Diese Anweisungen exportieren zunächst die Preisliste im RTF-Format, legen dann eine neue Nachricht an, weisen dieser Nachricht die zuvor generierte Preisliste sowie ein extern vorliegendes PDF-Dokument als Anhang zu und zeigen die Nachricht dann zur weiteren Bearbeitung an.

